
GoosManningSalomons_Table1.do
*************************************************************************
***********Makes Table 1 of "Employment Polarization in Europe***********
* analyzing changes in the job structure in Europe by	calculating the 	*
* change in occupational employment shares between 1993 and 2006, where	* 
* employed is pooled across the European countries.				*
* Since not all countries have data spanning the complete range of	*
* years between 1993-2006, we first impute employment data in 1993 	*
* and/or 2006 at the occupation-industry level for each country with 	*
* a shorter data span by using that country's average annual growth 	*
* rate of employment in the occ-ind cell. 					*
* We do this by taking the first and the last year in which the occ-ind *
* cell figures and calculating the average annual growth rate, 		*
* conditional on the observations being at least 3 years apart. We then *
* use this growth rate to impute employment in the ind-occ cell 		*
* in 1993 and/or 2006. We drop ind-occ cells where imputation could 	*
* not be done. [Note that results are not sensitive to this imputation	*
* method: not imputing data but instead relying on a shorter data span	*
* for all countries does not qualitatively affect results.]			*
* Having thus obtained a employment for 1993 and 2006 for all countries,*
* we collapse the data by occupation (21 in total, classified with ISCO)*
* From there we calculate the change in employment shares for each 	*
* occupation between 1993 and 2006.							*
*************************************************************************


*1. IMPUTING EMPLOYMENT IN 1993&2006 FOR MISSING OCC CELLS, BASED ON AVERAGE ANNUAL GROWTH RATES

set more off
cd E:\RESEARCH\EUpol\Tables
* annual_collapsed16.dta is the cleaned ELFS dataset collapsed to the country-occupation-industry-year level,
* appended with cleaned German IABS data at the same level.
* the variable coeff denotes persons employed, whereas hwcoeff denotes weekly hours worked
use annual_collapsed16.dta, clear
drop lnhwcoeff lncoeff coeff
sort country
egen ctry=group(country)
save temp_empshare, replace
local i=1
while `i'<=16 {
use temp_empshare, clear
keep if ctry==`i'
egen minyear=min(year)
egen maxyear=max(year)
replace year=1993 if year==minyear
replace year=2006 if year==maxyear
keep if year==1993|year==2006
reshape wide hwcoeff, i(occ ind) j(year)
collapse (sum) hwcoeff1993 hwcoeff2006 (mean) minyear maxyear, by(occ ind)
bysort occ ind: gen annualg=(hwcoeff2006-hwcoeff1993)/hwcoeff1993/(maxyear-minyear) if maxyear-minyear>2
bysort occ ind: gen emp93=hwcoeff1993/(1+annualg*(minyear-1993))
bysort occ ind: gen emp06=hwcoeff1993*(1+annualg*(2006-minyear))
drop hwcoeff*
*For a very small number of occ-ind cells, imputation based on average growth rates is unreliable: we drop these cells
drop if emp93==. | emp06==. | emp93<0 | emp06<0
keep occ ind emp93 emp06
gen ctry=`i'
save temp`i', replace
local i=`i'+1
}

*2. CALCULATING MEAN PERCENTAGE AND PERCENTAGE POINT CHANGES IN OCCUPATIONAL EMPLOYMENT SHARES, POOLING OCCUPATIONAL EMPLOYMENT ACROSS COUNTRIES

use temp1, clear
local i=2
while `i'<=16 {
append using temp`i'
local i=`i'+1
}
collapse (sum) emp93 emp06, by(occ)
egen totemp93=sum(emp93)
egen totemp06=sum(emp06)
bysort occ: gen empshare93=emp93/totemp93*100
bysort occ: gen empshare06=emp06/totemp06*100
bysort occ: gen dempshare=empshare06-empshare93
gen pdempshare=(empshare06-empshare93)/empshare93*100
keep occ empsh* demp pdemp




GoosManningSalomons_Table2.do
*************************************************************************
**********Makes Table 2 of "Employment Polarization in Europe",**********
* analyzing the pervasiveness of employment polarization in Europe by	*
* calculating the change in high-, middling, and low-paid occupational	*
* employment shares between 1993 and 2006 for each of the 16 European	*
* countries separately. 								*
* Since not all countries have data spanning the complete range of	*
* years between 1993-2006, we first impute employment data in 1993 	*
* and/or 2006 at the occupation-industry level for each country with 	*
* a shorter data span by using that country's average annual growth 	*
* rate of employment in the occ-ind cell. 					*
* We do this by taking the first and the last year in which the occ-ind *
* cell figures and calculating the average annual growth rate, 		*
* conditional on the observations being at least 3 years apart. We then *
* use this growth rate to impute employment in the ind-occ cell 		*
* in 1993 and/or 2006. We drop ind-occ cells where imputation could 	*
* not be done. [Note that results are not sensitive to this imputation	*
* method: not imputing data but instead relying on a shorter data span	*
* for all countries does not qualitatively affect results.]			*
* Having thus obtained a employment for 1993 and 2006 for all countries,*
* we collapse the data by country and occupation group, where an 		*
* occupation group is defined as 8 highest-paying, 9 middling, or 4	*
* lowest-paying occupations (as defined in Table 1). 				*
* From there we calculate the change in employment shares for each of 	*
* these 3 groups in each country between 1993 and 2006.			*
*************************************************************************


*1. IMPUTING EMPLOYMENT IN 1993&2006 FOR MISSING OCC CELLS, BASED ON AVERAGE ANNUAL GROWTH RATES

set more off
cd E:\RESEARCH\EUpol\Tables
* annual_collapsed16.dta is the cleaned ELFS dataset collapsed to the country-occupation-industry-year level,
* appended with cleaned German IABS data at the same level.
* the variable coeff denotes persons employed, whereas hwcoeff denotes weekly hours worked
use annual_collapsed16.dta, clear
drop lncoeff lnhwcoeff coeff
sort country
egen ctry=group(country)
save temp_empshare, replace
local i=1
while `i'<=16 {
use temp_empshare, clear
keep if ctry==`i'
bysort ind occ: egen minyear=min(year)
bysort ind occ: egen maxyear=max(year)
replace year=1993 if year==minyear
replace year=2006 if year==maxyear
keep if year==1993|year==2006
reshape wide hwcoeff, i(occ ind) j(year)
collapse (sum) hwcoeff1993 hwcoeff2006 (mean) minyear maxyear, by(occ ind)
bysort occ ind: gen annualg=(hwcoeff2006-hwcoeff1993)/hwcoeff1993/(maxyear-minyear) if maxyear-minyear>2
bysort occ ind: gen emp93=hwcoeff1993/(1+annualg*(minyear-1993))
bysort occ ind: gen emp06=hwcoeff1993*(1+annualg*(2006-minyear))
*For a very small number of occ-ind cells, imputation based on average growth rates is unreliable: we drop these cells
drop if emp93==. | emp06==. | emp93<0 | emp06<0
keep occ ind emp93 emp06
gen ctry=`i'
save temp`i', replace
local i=`i'+1
}

*2. CALCULATING MEAN PERCENTAGE AND PERCENTAGE POINT CHANGES IN COUNTRIES' HIGH-, MIDDLE-, & LOW-PAID OCCUPATIONAL EMPLOYMENT SHARES

use temp1, clear
local i=2
while `i'<=16 {
append using temp`i'
local i=`i'+1
}
*collapsing employment into 8 highest-, 9 middling, and 4 lowest-paid occs
*the numbers below refer to occupations' 2-digit ISCO codes
replace occ=1 if occ==91|occ==93|occ==52|occ==51
replace occ=2 if occ==41|occ==73|occ==72|occ==83|occ==81|occ==42|occ==71|occ==82|occ==74
replace occ=3 if occ==12|occ==21|occ==22|occ==24|occ==13|occ==31|occ==34|occ==32
collapse (sum) emp93 emp06, by(ctry occ)
*calculating changes in employment shares of the three groups of occupations for each country
bysort ctry: egen totemp93=sum(emp93)
bysort ctry: egen totemp06=sum(emp06)
gen empshare93=emp93/totemp93*100
gen empshare06=emp06/totemp06*100
gen dempshare=empshare06-empshare93
gen pdempshare=dempshare/empshare93*100
keep empshare93 demp pdemp ctry occ
reshape wide empshare93 demp pdemp, i(ctry) j(occ)




GoosManningSalomons_Table3.do
*************************************************************************************
***************Makes Table 3 of "Employment Polarization in Europe",*****************
* regressing the log of employment in occupation-country-year cells onto occupation-*
* specific task measures, offshorability and education, all interacted with a 	*
* linear time trend; controlling for the country-occupation-year specific log wage	*
* and countr-occupation and country-year dummies.						*
* We standardize task measures, offshorability and the education level to make	*
* their impacts comparable. Standard errors are clustered at the country-occupation *
* level to allow for serial correlation across years.						*
*************************************************************************************

set more off
cd E:\RESEARCH\Eupol\Tables
*This dataset contains all our data, see read_me.pdf
use Eupol_alldata, clear

*Standardizing task measures (R=routine, A=Abstract, S=Service), offshorability (OFF), and education (E) to make their impacts comparable
egen newvar=std(R)
drop R
rename newvar R
egen newvar=std(A)
drop A
rename newvar A
egen newvar=std(S)
drop S
rename newvar S
egen newvar=std(OFF)
drop OFF
rename newvar OFF
rename educ E
egen newvar=std(E)
drop E
rename newvar E

*Generating dummies
egen jc=group(occupation country)
egen ct=group(country year)

*Generating linear time trend variable
generate tt=1
replace tt=0 if year==1993
replace tt=1 if year==1994
replace tt=2 if year==1995
replace tt=3 if year==1996
replace tt=4 if year==1997
replace tt=5 if year==1998
replace tt=6 if year==1999
replace tt=7 if year==2000
replace tt=8 if year==2001
replace tt=9 if year==2002
replace tt=10 if year==2003
replace tt=11 if year==2004
replace tt=12 if year==2005
replace tt=13 if year==2006

*Interacting task measures, offshorability and the education level with the time trend
gen tt_A=tt*A
gen tt_R=tt*R
gen tt_S=tt*S
gen tt_O=tt*OFF
gen tt_E=tt*E

*Running the regression (note that lwh is the log wage), clustering std errors at the occupation-country level
xi:reg lnhwcoeff tt_A tt_R tt_S tt_E tt_O lwh i.ct i.jc, vce(cluster jc)

*Dropping variables with insignificant point estimates
xi:reg lnhwcoeff tt_A tt_R i.ct i.jc, vce(cluster jc)

*Not reported as table, but mentioned in text: testing for country-heterogeneity in the impacts of technological change and offshoring
*by interacting task_measure*timetrend and offshorability*time_trend with country dummies and testing for joint significance
char country[omit]"austria"
xi:reg lnhwcoeff i.country*tt_A i.country*tt_R i.country*tt_S I.country*tt_O i.country*A i.country*R i.country*S i.country*O lwh i.jc i.ct, vce(cluster jc)
testparm _IcouXtt_A*
testparm _IcouXtt_R*
testparm _IcouXtt_S*
testparm _IcouXtt_O*




GoosManningSalomons_Figure1.do
*************************************************************************************
***************Makes Figure 1 of "Employment Polarization in Europe",****************
* plotting percentage changes in employment shares of jobs ranked by their log wage.*
* Job is defined as occ-ind cell, and the wage is the UK 1994 occ-ind log wage.	*
* Data are pooled across the European countries.						*
* We again impute employment data in 1993 and/or 2006 for countries where these 	*
* years are not available (see do-file Table1 for description of imputation method).*
*************************************************************************************


*1. IMPUTING EMPLOYMENT IN 1993&2006 FOR MISSING OCC CELLS, BASED ON AVERAGE ANNUAL GROWTH RATES

set more off
cd E:\RESEARCH\EUpol\Tables
* annual_collapsed16.dta is the cleaned ELFS dataset collapsed to the country-occupation-industry-year level,
* appended with cleaned German IABS data at the same level.
* the variable coeff denotes persons employed, whereas hwcoeff denotes weekly hours worked
* we use this dataset rather than Eupol_alldata.dta, which contains all data, because in the latter 
* employment is collapsed at the country-occ-year level but we use the industry dimension for our imputations
use annual_collapsed16.dta, clear
drop lnhwcoeff lncoeff coeff
sort country
egen ctry=group(country)
save temp_empshare, replace
local i=1
while `i'<=16 {
use temp_empshare, clear
keep if ctry==`i'
egen minyear=min(year)
egen maxyear=max(year)
replace year=1993 if year==minyear
replace year=2006 if year==maxyear
keep if year==1993|year==2006
reshape wide hwcoeff, i(occ ind) j(year)
collapse (sum) hwcoeff1993 hwcoeff2006 (mean) minyear maxyear, by(occ ind)
bysort occ ind: gen annualg=(hwcoeff2006-hwcoeff1993)/hwcoeff1993/(maxyear-minyear) if maxyear-minyear>2
bysort occ ind: gen emp93=hwcoeff1993/(1+annualg*(minyear-1993))
bysort occ ind: gen emp06=hwcoeff1993*(1+annualg*(2006-minyear))
drop hwcoeff*
*For a very small number of occ-ind cells, imputation based on average growth rates is unreliable: we drop these cells
drop if emp93==. | emp06==. | emp93<0 | emp06<0
keep occ ind emp93 emp06
gen ctry=`i'
save temp`i', replace
local i=`i'+1
}


*2. CALCULATING PERCENTAGE CHANGE IN OCC-IND EMPLOYMENT SHARES

use temp1, clear
local i=2
while `i'<=16 {
append using temp`i'
local i=`i'+1
}
collapse (sum)  emp93 emp06, by(ind occ)
egen totemp93=sum(emp93)
egen totemp06=sum(emp06)
bysort occ ind: gen empshare93=emp93/totemp93*100
bysort occ ind: gen empshare06=emp06/totemp06*100
bysort occ ind: gen dempshare=empshare06-empshare93
keep occ ind empshare93 empshare06
gen pdempshare=(empshare06-empshare93)/empshare93*100
drop if pdempshare>100|pdempshare<-100
order occ ind empshare93 empshare06 pdempshare
sort occ ind
save temp_dempshares.dta, replace


*3. MERGING IN THE UK OCC-IND WAGES

cd E:\RESEARCH\Eupol\LFS
*this dataset contains, among others, 1994 UK wage data at the occ-ind level
use annual_collapsed+w.dta, clear
keep if country=="uk"&year==1994
keep occ ind lw
sort occ ind
cd E:\RESEARCH\EUpol\Tables 
merge occ ind using temp_dempshares.dta
keep if _==3
drop _
save temp_figure1.dta, replace


*4. MAKING FIG 1

use temp_figure1.dta, clear
*we trim the outliers (approximately the 1st & 99th percentiles of the occ-ind wage distribution)
drop if lw<1.434344|lw>=2.663506 
kernreg  pdempshare lw, b(0.25) k(3) np(250) g(mhvar gridvar)
keep mhvar gridvar
drop if mhvar==.
rename gridvar lw
save temp, replace

use temp_figure1.dta, clear
drop if lw<1.434344|lw>=2.663506
append using temp
replace empshare93=1 if mhvar~=.
label var lw "UK log mean wage, 94"
label var pdempshare "% change in employment share 93-06"
label var mhvar "fitted kernel"


*Figure 1 fitting AER p&p guidelines  

two (scatter pdempshare lw [w=empshare93], sort graphregion(margin(medsmall)) legend(off) xsize(2.625) ysize(2) xlabel(#3, labsize(large)) ylabel(-100(50)100, nogrid labsize(large)) /*
*/xtitle("Log mean wage in 1994", size(large)) ytitle("Change in employment share (percent)", size(large)) msymbol(Oh) mcolor(black) graphregion(fcolor(white)) plotregion(lcolor(black))/*
*/yline(0, lcolor(black))) (line mhvar lw, sort clwidth(thick) clcolor(black))

*Figure 1 used in presentation (same content but different look)

two (scatter pdempshare lw [w=empshare93], sort scheme(s1mono) ylabel(-100(50)100, nogrid) xlabel(#3) ytitle("% change in share of hours worked(93-06)")/*
*/xtitle("Log mean wage") msymbol(Oh) mcolor(black) plotregion(lcolor(black))  title("Figure 1. European-wide Job Polarization, 1993-2006") yline(0))/*
*/(line mhvar lw, sort clwidth(thick) clcolor(black))





GoosManningSalomons_Figure2.do
*************************************************************************************
***************Makes Figure 2 of "Employment Polarization in Europe",****************
* plotting employment shares of low-wage jobs in 1993 against countries' overall	*
* wage inequality (measured as the log of p90/p10).						*
* We again impute employment data in 1993 for countries where this year is not 	*
* available (see the do-file of Table 1 for a description of the imputation method).*
*************************************************************************************


*1. IMPUTING EMPLOYMENT IN 1993 FOR MISSING OCC CELLS, BASED ON AVERAGE ANNUAL GROWTH RATES*/
set more off
cd E:\RESEARCH\EUpol\Tables
use annual_collapsed16.dta, clear
drop lnhwcoeff lncoeff
drop if hwcoeff==.
sort country
egen ctry=group(country)
save temp_empshare, replace
local i=1
while `i'<=16 {
use temp_empshare, clear
keep if ctry==`i'
drop coeff
bysort ind occ: egen minyear=min(year)
bysort ind occ: egen maxyear=max(year)
replace year=1993 if year==minyear
replace year=2006 if year==maxyear
keep if year==1993|year==2006
reshape wide hwcoeff, i(occ ind) j(year)
collapse (sum) hwcoeff1993 hwcoeff2006 (mean) minyear maxyear, by(occ ind)
bysort occ ind: gen annualg=(hwcoeff2006-hwcoeff1993)/hwcoeff1993/(maxyear-minyear) if maxyear-minyear>2
bysort occ ind: gen emp93=hwcoeff1993/(1+annualg*(minyear-1993))
drop if emp93<0 | emp93==. 
keep occ ind emp93 
gen ctry=`i'
save temp`i', replace
local i=`i'+1
}
use temp1, clear
local i=2
while `i'<=16 {
append using temp`i'
local i=`i'+1
}

*2. CALCULATING PERCENTAGE CHANGE IN EMPLOYMENT SHARE OF $ LOWEST-PAID OCCUPATIONS

replace occ=1 if occ==91|occ==93|occ==52|occ==51
replace occ=0 if occ~=1
collapse (sum) emp*, by(ctry occ)
gen country="austria" if ctry==1
replace country="belgium" if ctry==2
replace country="denmark" if ctry==3
replace country="finland" if ctry==4
replace country="france" if ctry==5
replace country="germany" if ctry==6
replace country="greece" if ctry==7
replace country="ireland" if ctry==8
replace country="italy" if ctry==9
replace country="luxembourg" if ctry==10
replace country="netherlands" if ctry==11
replace country="norway" if ctry==12
replace country="portugal" if ctry==13
replace country="spain" if ctry==14
replace country="sweden" if ctry==15
replace country="uk" if ctry==16
bysort country: egen totemp=sum(emp)
bysort country: gen empshare=emp93/totemp*100
keep country occ empshare
reshape wide empshare, i(country) j(occ)
sort country

*3. MERGING IN THE WAGE INEQUALITY DATA

cd E:\RESEARCH\Eupol\Tables\
merge country using wage_inequality
drop _

*4. MAKING FIG 2

gen ccode="at" if country=="austria"
replace ccode="be" if country=="belgium"
replace ccode="dk" if country=="denmark"
replace ccode="fi" if country=="finland"
replace ccode="fr" if country=="france"
replace ccode="de" if country=="germany"
replace ccode="gr" if country=="greece"
replace ccode="ie" if country=="ireland"
replace ccode="it" if country=="italy"
replace ccode="lu" if country=="luxembourg"
replace ccode="nl" if country=="netherlands"
replace ccode="no" if country=="norway"
replace ccode="pt" if country=="portugal"
replace ccode="es" if country=="spain"
replace ccode="sw" if country=="sweden"
replace ccode="uk" if country=="uk"
label var empshare1 "Employment share of low-wage occupations in 1993"
label var lp90lp10 "log(p90/p10) averaged across 1993-2006"
*generating a variable that specifies where the labels of these datapoints have to be located in Figure 2
gen labpos=9
replace labpos=12 if ccode=="at"
replace labpos=12 if ccode=="fr"
replace labpos=6 if ccode=="gr"
replace labpos=6 if ccode=="lu"
replace labpos=6 if ccode=="sw"
*getting fitted values
reg empshare1 lp90lp10
predict hat

*Figure 2 fitting AER p&p guidelines  

twoway (scatter empshare1 lp90lp10, sort xtitle("Log(p90/p10)", size(large)) ytitle("Low-wage employment share in 1993", size(large))/*
*/legend(off) xsize(2.625) ysize(2) xlabel(#5,format(%9.1fc)labsize(large)) ylabel(#3, nogrid labsize(large)) graphregion(margin(medsmall))/* 
*/mcolor(gray) mlabel(ccode) mlabgap(1.5) mlabsize(large) mlabvposition(labpos) graphregion(fcolor(white)) plotregion(lcolor(black)))/*
*/(line hat lp90lp10, blpattern(solid) scheme(s2mono) clwidth(thick) clcolor(black))

*Figure 2 used in presentation (same content but different look)

twoway (scatter empshare1 lp90lp10, sort title("Figure 3. Overall Wage Inequality and the 1993" "Low-Wage Employment Share in 16 European countries")/*
*/xtitle("Log(p90/p10)", size(large)) ytitle("Low-wage employment share in 1993", size(large)) xlabel(#5,format(%9.1fc)) ylabel(#3, nogrid)/*
*/mlabel(ccode) mcolor(gray) mlabgap(1.5) mlabsize(large) mlabvposition(labpos) graphregion(fcolor(white)) plotregion(lcolor(black)) legend(off))/*
*/(line hat lp90lp10, blpattern(solid) scheme(s2mono) clwidth(thick) clcolor(black))

Table2.do  

These files are all Stata10 do-files. Descriptions of how the tables and figures are 


made can be found in the dofiles themselves, in boxes as follows: 


************ 
*   this that * 
************ 
 
The following datasets are used:  


1. European Union Labour Force Survey (ELFS) [employment data for 15 


European countries; occupations’ average education level] 


2. IABS [employment data for Germany] 


3. United Kingdom Labour Force Survey (UKLFS) [industry-occupation specific 


wage data; only used in Figure 1] 


4. Occupational Information Network dataset (ONET) [occupation-specific 


measure of tasks affected by technological change] 


5. European Restructuring Monitor (ERM) of the European Monitoring Centre on 


Change (EMCC), which is a part of the European Foundation for the 







Improvement of Living and Working Conditions (Eurofound) [occupation-


specific measure of tasks affected by offshoring] 


6. European Union Statistics on Income and Living Conditions (EU-SILC) and its 


predecessor the European Community Household Panel (ECHP) [wages at the 


country-occupation-year level] 


 


Below are descriptions of how to obtain each of these datasets, and how we used 


them in our paper. 


 


ELFS 


How to access 


The ELFS can be obtained from Eurostat. Application procedures differ for 


organizations inside the EU and outside the EU. Contact:  


estat-microdataaccess@ec.europa.eu for EU  


martine.peeters@ec.europa.eu for non-EU 


The full dataset costs EUR 8,000- parts of the data (e.g. for a selection of countries 


and/or years) can be obtained at lower cost. More information for those interested in 


buying the data can be found at: 


http://epp.eurostat.ec.europa.eu/pls/portal/docs/PAGE/PGP_DS_MICRODATA/PGE_D


S_MICRODATA_01/TAB47568350/EN%20-%20LFS-


PDF%20PART%20CONVERSION%20OF%20ORIGINAL%20DOC%20%20-


%20WEBSITE%20-%20ACCESS%20TO%20MICRODATA.PDF  


Dataset 







The ELFS contains data for 29 European countries over 1983-20061 which is collected 


on a national level. The same set of characteristics is recorded in each country, 


common classifications and definitions are used, and data processed centrally by 


Eurostat. We limit our analyses to the fifteen countries that made up the European 


Union previous to the 2004 enlargement, plus Norway and minus Germany (the 


latter due to poor data quality).  


We merged the annual datasets together for each of the 15 countries and cleaned 


these datasets by retaining only the employed (using the ILO definition of 


employment, embodied in the variable “ilostat”), and dropping the employed for 


whom no (NACE major group) industry and (2-digit ISCO) occupation codes are 


available. We also drop the employed reporting no usual weekly hours worked 


(variable “hwusual”) or reporting to work zero hours. Finally, we drop several years 


for two countries where an incomplete panel of industries or occupations is 


represented (specifically, Ireland 1993-1997 and Italy 2000-2003). We aggregate 


the individual weighting factor (a variable called “coeff”) up to the country-


occupation-industry-year level, and multiply it by usual hours worked to obtain an 


hours-weighted measure of employment (- we call this variable hwcoeff).  


The ELFS is also the source of our education information. For this, we use a three-


level education variable (hatlev1d) classified with ISCED: the lowest level of 


education corresponds to ISCED 0, 1, and 2 (pre-primary education; primary and 


lower secondary education); the middle level to ISCED 3 and 4 (upper secondary and 


pos-secondary non-tertiary education); and the highest level to ISCED 5 and 6 


(tertiary and postgraduate education).  


Note that the ELFS does not yet contain wage information: for this, we rely on the 


ECHP and EU-SILC (see below). 


                                                 
1 This is the maximum time span. Some countries enter the ELFS later, and generally speaking, the new 
EU members only enter the sample in the year they entered the EU. We only use data from 1993 onwards 
since there is not occupation and industry information available in previous years. 







 


IABS 


How to access 


The IABS can be obtained from the Research Data Centre of the German Federal 


Employment Agency at the Institute for Employment Research 


(Forschungsdatenzentrum - Institut für Arbeitsmarkt- und Berufsforschung), 


http://fdz.iab.de. A research proposal has to be submitted to iab.fdz@iab.de, 


together with information about how the data will be used and protected. Once the 


proposal is approved, the data are free of charge, but cannot be freely distributed.  


See http://fdz.iab.de/en/FDZ_Data_Access/FDZ_Scientific_Use_Files.aspx and 


http://fdz.iab.de/en/FDZ_Individual_Data/IAB_Employment_Samples.aspx for more 


information.  


Dataset 


The IABS is a 2% random sample of German social security records covering 1993-


2004. Since the 2-digit occupation and industry codes used in the IABS differ 


somewhat from ISCO and NACE and no crosswalk was available, we matched them 


manually. Due to anonymization, occupation and industry codes in the IABS are not 


more disaggregate than the ones in the ELFS, and as a result we were not able to 


find a match for 2 ISCO occupations and 4 NACE industries. Instead, German 


employment in these occupations and industries is included in other ISCO and NACE 


categories: however, none of our analyses are sensitive to the exclusion of Germany. 


Lastly, the IABS industry classification changes in 2003: this classification is 


somewhat easier to reconcile with NACE, but since it covers only 2 years and no 


crosswalk exists between the IABS industry classifications before and after 2003, we 


drop years 2003 and 2004. Since some government workers are not liable to social 


security, the data is not random for them. We drop these occupations from the 







sample, as well as from the ELFS dataset to keep comparability- although results are 


not sensitive to this.  


The IABS contains information on full-time / part-time status and region of work 


(West- or East-Germany) but not on hours worked. We calculate hours worked by 


combining the full-time / part-time information with time-varying information on 


average weekly hours worked for full-time and part-time workers in West- and East-


Germany, obtained from Eurofound. 


 


UKLFS 


How to access 


The UK LFS can be obtained from the UK Data Archive http://www.data-


archive.ac.uk/. Users from outside the UK will have to fill in an Athens registration 


form. More information on the data: 


http://www.statistics.gov.uk/StatBase/Source.asp?vlnk=358&More=Y. 


More information about accessing the LFS: 


http://www.data-archive.ac.uk/aandp/access/login.asp  


Dataset 


We only use wage information from the UKLFS since the employment information is 


already contained in the ELFS. We collapse gross weekly wages to the occupation-


industry-year level, using the provided sample weights. Since the number of 


observations in 1993 is significantly lower than in subsequent years, we rely on 


occupation-industry wage rankings from 1994 to make Figure 1.  


 


ONET 


How to access 


The ONET database is developed under the sponsorship of the US Department of 


Labor and can downloaded from http://www.onetcenter.org/. 







Dataset 


Version 11 of the ONET database (released in 2006) contains 161 US occupation-


specific time-invariant variables (ordered within a so-called content model), many of 


which can be seen as representing tasks people do on the job. We use 96 variables 


from 5 different sections: from Worker Characteristics, we use Abilities (section 1A), 


from Worker Requirements, we use Basic Skills and Cross-Functional Skills (sections 


2A and 2B), and from Occupational Requirements, we use Generalized Work 


Activities and Work Context (sections 4A and 4B). All 96 variables have the 


importance scale, where the respondent and/or occupational expert ranks each task 


as not important at all (1), somewhat important (2), important (3), very important 


(4) or extremely important (5). We categorized the variables into one of three tasks 


(Abstract, Routine or Service) based on the ALM-hypothesis of how well technology 


can substitute for these tasks- from this we calculate 3 principal components. We 


then collapse the principal components to the ISCO level by weighting them by their 


US occupational employment size in 2005, which we obtain from the Bureau of Labor 


Statistics. We now have a dataset with ONET task measures at the ISCO level. We 


rescale the three task measures such that they have a zero mean and unit standard 


deviation.  


 


ERM 


How to access 


The news reports about international outsourcing (“offshoring”) can be downloaded 


from http://www.eurofound.europa.eu/emcc/index.htm.  


Dataset 


The European Restructuring Monitor (ERM) contains summaries of news reports about 


cases of offshoring by companies located in Europe. Started in May 2002, 460 reports 


were available up to June 20th, 2008. From these news reports, called fact sheets, we 







abstracted information about the occupations that were being offshored. This assigning 


of occupations was relatively straightforward in most cases, both because the reports 


are quite extensive and because our occupational classification is very aggregated. 


Whenever it was not possible to deduce the offshored occupation(s) from the fact sheet, 


we turned to the original news report provided in the fact sheet, and if that was not 


sufficient, looked on the company’s website. Maximizing information in this way, we 


were able to obtain offshored occupations for 415 of the 460 fact sheets. We then count 


the number of cases by ISCO occupation as a measure for that occupation’s 


offshorability and from that generate an offshorability rank with mean zero and unit 


standard deviation across occupations. 


 


ECHP 


The ECHP can be obtained from Eurostat. Application procedures differ for 


organizations inside the EU and outside the EU. Contact:  


estat-microdataaccess@ec.europa.eu for EU  


martine.peeters@ec.europa.eu for non-EU 


The cost is EUR 250 for the entire dataset. More information for those interested in 


buying the data can be found at: 


http://epp.eurostat.ec.europa.eu/pls/portal/docs/PAGE/PGP_DS_MICRODATA/PGE_D


S_MICRODATA_01/TAB47568297/EN-ECHP-WORD-PDF%20CONVERSION%20-


%20WEBSITE%20-%20ACCESS%20TO%20MICRODATA%20-%20EN.PDF 


Dataset 


We use data on gross monthly wages in national currencies over 1993-2001 for 


European countries: we convert them into 2000 Euros using harmonized price indices 


and 2000 real exchange rates. Wages are then collapsed to the occupation-country-


year level using the provided sample weights which have been premultiplied by 


hours worked.  







 


EU-SILC 


How to access 


The EU-SILC can be obtained from Eurostat. Application procedures differ for 


organizations inside the EU and outside the EU. Contact:  


estat-microdataaccess@ec.europa.eu for EU  


martine.peeters@ec.europa.eu for non-EU 


The cost is EUR 1,000 for the full dataset. More information for those interested in 


buying the data can be found at: 


http://epp.eurostat.ec.europa.eu/pls/portal/docs/PAGE/PGP_DS_MICRODATA/PGE_D


S_MICRODATA_01/EU-SILCTAB/EN-EU-SILC-WEBSITE-PDF%20PART.PDF  


Dataset 


Contains gross monthly wages in Euros over 2004-2006 for European countries. We 


convert them into 2000 Euros using harmonized price indices and 2000 real exchange 


rates. Wages are then collapsed to the occupation-country-year level using the 


provided sample weights which have been premultiplied by hours worked.  


Notes on wage data:  


- Taking ECHP and EU-SILC data together, we do not yet have country-specific 


occupational wages for each year. Given that we find that the occupational 


wage ranking is very stable within countries over time (Spearman rank 


correlation coefficients of around 0.90, significant at the 1% level), we impute 


wages for missing years by setting them equal to the average wage across the 


closest years where original data is available.  


The ECHP and EU-SILC do not contain wage data for Finland and Sweden. We 


therefore use the following method to impute occupation-year specific wages in 


these two countries: ( )tDEtDEj

where jctw  is the average wage in occupation j, in country c (in this case, Finland 


or Sweden) at time t, ctw  is the median wage in country c at time t, and ctσ  is a 


measure of wage inequality in country c at time t (specifically the ratio of the 90th 


to the 10th percentile derived from the OECD). The variables with the subscript DE 


refer to the value of those variables in Germany.  


- In each country wages have been smoothed twice. First, by pooling together all 


years for each occupation and estimating a model in which the dummy on 


occupation varies smoothly with a quadratic time trend. Secondly, by taking the 


fitted values from a regression of the smoothed wages onto country-year 


dummies, occupation-country dummies and occupation-country dummies 


interacted with a linear time trend. 


 


Combining the employment data (ELFS, IABS) at the country-occupation-year level; 


education data (ELFS, IABS) at the occupation level; wage data (UKLFS, ECHP, EU-


SILC) at the country-occupation-year level; measures of technological change 


(ONET); and offshoring (ERM) at the occupation level we obtain the dataset (called 


Eupol_alldata) used in the do-files contained in this folder. 


 






